



296: Everyday OOABL

Oct 7, 2019

Chad R. Thomson

Senior Principal Consultant


Progress Software

Speaker BIO

- Over 20 years of industry experience, favoring reality over formality
- Specializing in vendor-neutral, cross-platform application and service integration
- Passionate technology advocate
- Session code examples and slide-updates available in GitHub repo:
 - <https://github.com/ChadThomsonPSC/pugna-2019-everyday-oop>




Only those who have the patience to do simple things perfectly will acquire the skill to do difficult things easily



Agenda

- Why Bother with OOABL?
- Setting up for Success
- OOABL Primer
- Procedural vs OOABL Comparison
- Every-day Use Examples
- Next Steps




Why Bother with OOABL?


- OOABL will enforce method and type checking during **compilation**
 - Doesn't fix code/logic, but mitigates opportunity for failure
 - Even the most experienced developers still encounter run-time errors when procedure signatures don't match
- You want to **modernize**, right?
 - PASOE web handlers, (REST) BusinessEntity – Class-based
- Object caching and memory management, superior to procedural counter-parts
 - Garbage collection assists in *limiting* the amount of clean-up concerns
 - Please, pick-up after yourself, though
- Inheritance and static classes can be simple scaffolding for application architecture
 - reduces the amount of traditional "framework" management code
 - eg. super-procedure-based: run, persistent, make super, handle locate, repeat

Why Bother: ...it's too difficult


- It doesn't have to be
 - It can be as simple or complex as you want or need it to be
- Don't get caught-up in the **academics** of it all
 - Design Patterns
 - Static Singleton Factory Provider Pattern
 - Inversion of Control, Dependency Injection
 - Polymorphism
 - Wait, what?
- Start with small, simple tasks
 - Global functions
 - Logging
 - Messaging
- Slowly **mix-in** OOABL to your procedural code



Setting Up for Success




Setting Up for Success: the IDE




- Use **PDSOE** for projects
 - Convenience libraries usually already included in PROPATH
 - OpenEdge.Core.pl
 - OpenEdge.BusinessLogic.pl
 - OpenEdge.Net.pl

Setting Up for Success: Get the Source, View the Source

- Get the **ADE Dev Tools** source
- <https://community.progress.com>
- Link in ADE Dev Tools source
- Use the Class Browser



OOABL Primer



OOABL Primer: Anatomy of a Class

- **using === OOP Propath**
 - @ line 2,3
 - Replace "." with "/" to find Class r-code
 - Exception "Progress.*"
- **Class "Package" or "Namespace"**
 - @ line 8
 - Means: path to folder containing class files
- **Inherits "Progress.Lang.Object" by default**
 - @ line 9

```
StandardExample.cls
1  /* USING === OOP propath */
2  using Progress.Lang.*.
3  using Progress.Lang.Object.
4
5  /* "augment" error handling */
6  block-level on error undo, throw.
7
8  class pca.StandardExample
9      inherits Progress.Lang.Object /* redundant */
10     :
11     /* DataSet, TempTable includes go here */
12
13     /* Properties */
14     define public property myProperty as char no-undo
15     get:
16         if (myProperty eq ? or
17             myProperty eq "" ) then
18             assign myProperty = "Hello".
19
20         return myProperty.
21     end.
22     private set.
23
24     /* Default Constructor*/
25     constructor public StandardExample ( ):
26         super ().
27     end constructor. /* standardExample */
28
29     /* Public Methods */
30     method public void ExampleMethod( ):
31         return.
32     end method.
33
34     /* Default Destructor*/
35     destructor public StandardExample ( ):
36     end destructor.
37
38 end class.
39 /* EOF */
```


OOABL Primer: Anatomy of a Class

- Variables, Datasets, etc
 - @ line 11
 - Called: **[Data-]Members**
- **Properties**
 - @ line 14
 - Variables with built-in logic
- **Constructor, Destructor**
 - @ line 25
 - Optional*
 - * *based-on super-class*
- Member Accessibility modes
 - @ line 30
 - **Public, Private, Protected**

```
StandardExample.cls
1  /* USING === OOP propath */
2  using Progress.Lang.*.
3  using Progress.Lang.Object.
4
5  /* "augment" error handling */
6  block-level on error undo, throw.
7
8  class pca.StandardExample
9      inherits Progress.Lang.Object /* redundant */
10     :
11     /* DataSet, TempTable includes go here */
12
13     /* Properties */
14     define public property myProperty as char no-undo
15     get:
16         if (myProperty eq ? or
17             myProperty eq "" ) then
18             assign myProperty = "Hello".
19
20         return myProperty.
21     end.
22     private set.
23
24     /* Default Constructor*/
25     constructor public StandardExample ( ):
26         super ().
27     end constructor. /* StandardExample */
28
29     /* Public Methods */
30     method public void ExampleMethod( ):
31         return.
32     end method.
33
34     /* Default Destructor*/
35     destructor public StandardExample ( ):
36     end destructor.
37
38 end class.
39 /* EOF */
```

OOABL Primer: Error Handling – Catch, Handle, and Throw

- Very convenient to start using **THROW, CATCH**
 - There is no Try, only blocks that THROW vs raise errors
 - Fewer lines of 'if .. then ... else'-style error processing logic
- Change your *mindset*
 - let errors happen, and handle vs test for them
- **FINALLY**
 - Runs as the very last line of executable code prior to procedure or block exiting completely
 - A good place for 'clean-up' logic
- Use of these error handling methods is a **requirement** when using Assertions (later)



OOABL Primer: Error Handling Mix-in

■ Throw

- Added to blocks
- Creates error object

■ Catch


- Inside blocks that Throw
- Passed error object for handling

■ Finally

- LAST block to execute
 - Even after 'return' statement
 - Can be inside other blocks
- Convenient place to clean-up objects, handles, etc

```
example1.p
1
2 using OpenEdge.Core.Assert from propath.
3
4 define variable sampleVar as character no-undo.
5 /* typical include file area */
6 /* {inc/thisfile.i} */
7
8 mainblock:
9 do on error undo, throw /* use the "throw" keyword to raise an error */
10 : /* for "catch" to address */
11
12 /* typical logic area */
13
14 /* raise a runtime error */
15 find Customer no-lock
16 no-error.
17
18 /* without "no-error" can assume customer is available here */
19 /* OR Assert that it's available */
20 Assert:IsAvailable(buffer Customer:handle, 'Customer').
21
22 catch err as Progress.Lang.Error
23 :
24 message "An error occurred"
25 skip err:GetClass():TypeName
26 skip err:GetMessage(1)
27 view-as alert-box.
28
29 delete object err.
30 assign err = ?.
31 end catch.
32 end. /* mainblock */
33
34 /* execution will continue to here if the error is handled */
35 /* and no new ones are thrown */
36 return.
37
38 finally:
39 /* code here will always be the *last* thing to run */
40 /* great place to clear/clean values and objects */
41 end finally.
42 /* EOF*/
```

Procedural vs OOABL Comparison



Procedural vs OOABL Comparison: Include vs Inherits

Procedural Include

```
01_include.i
1
2 /*-----
3 File      : 01_include.i
4 -----*/
5
6 /* ***** Definitions ***** */
7 define variable thisvariable as character no-undo.
8
9 function doSomething returns char
10 ( input withthisint as int ):
11
12     return string(withthisint).
13 end.
14 /* EOF*/

01_include.p
1
2 /*-----
3 File      : 01_include.p
4 -----*/
5
6 /* ***** Definitions ***** */
7
8 define variable sampleVar as character no-undo.
9
10 {include/01_include.i}
11
12 /* ***** Main Block ***** */
13 mainblock:
14 do:
15     assign
16     sampleVar = doSomething(1)
17     .
18
19     message sampleVar
20     view-as alert-box.
21 end. /* mainblock */
```

OOABL Inherits

```
IncludeExampleBase.cs
1 /*-----
2 File      : IncludeExampleBase
3 -----*/
4 using Progress.Lang.*.
5
6 block-level on error undo, throw.
7
8 class pca.IncludeExampleBase:
9     define public property thisvariable as char no-undo
10     get.
11     set.
12
13     method public char doSomething( input withthisint as int).
14     return string(withthisint).
15     end. /* doSomething */
16 end class.
17 /* EOF */

IncludeExample.cs
1 /*-----
2 File      : IncludeExample
3 -----*/
4 using Progress.Lang.*.
5 using pca.IncludeExampleBase.
6
7 block-level on error undo, throw.
8
9 class pca.IncludeExample
10     inherits IncludeExampleBase
11     :
12     constructor public IncludeExample ( ):
13     super ().
14
15     assign
16     thisvariable = doSomething(1)
17     .
18
19     message thisvariable
20     view-as alert-box.
21     end constructor.
22 end class.
23 /* EOF */
```


Procedural vs OOABL Comparison: Run it!

■ Procedural

- Use "run" statement

```
run_IncludeExampleProc.p
1
2 /*-----
3   File      : run_IncludeExampleProc.p
4   -----
5
6 /* ***** Definitions *****
7
8 block-level on error undo, throw.
9
10 /* ***** Main Block *****
11 run 01_include.p.
12
13 /* EOF */
```

Message (Press HELP to view stack trace)

1

OK Help

■ OOABL

- create a **NEW** "instance"
- Be sure to delete your object
- A more Apples-to-apples example
 - Better to call "logic" in a method
 - eg: "<charvar> = example01:doSomething(1)"


```
run_IncludeExampleClass.p
2 /*-----
3   File      : run_IncludeExampleClass.p
4   -----
5
6 /* ***** Definitions *****
7
8 block-level on error undo, throw.
9
10 define variable example01 as pca.IncludeExample no-undo.
11
12 /* ***** Main Block *****
13
14 /* Constructor will fire and
15  * run the 'doSomething' method
16  */
17 assign
18   example01 = new pca.IncludeExample()
19   .
20
21 /* clean-up */
22 delete object example01. /* destructor will fire */
23 assign
24   example01 = ?
25   .
26 /* EOF */
```


Procedural vs OOABL Comparison: Super vs Static

Procedural Super

```
superlib01.p
1 /*-----*/
2 File      : superlib01.p
3 /*-----*/
4
5 /* ***** Internal Procedures ***** */
6 procedure DoSomethingSuper:
7   /* logic here */
8 end. /* DoSomethingSuper */
9
10 function SystemDateTime returns char
11   ():
12   return string(datetime-tz(now)).
13 end.
14 /* EOF */


run_SuperAsProc.p
7 define variable superhdl as handle no-undo.
8 define variable systemdt as character no-undo.
9 /* ***** Main Block ***** */
10
11 /* function prototype OR use dynamic-function ({fn}) below */
12 function SystemDateTime returns char () in super.
13
14 run lib/superlib01.p
15 persistent
16 set superhdl
17 no-error.
18
19 if valid-handle(superhdl) then
20 do:
21   session:add-super-procedure (superhdl).
22
23   assign
24     systemdt = SystemDateTime()
25     /* systemdt = dynamic-function ('SystemDateTime')*/
26     /* systemdt = {fn "SystemDateTime"}*/
27     .
28
29   message subst('System DateTime is: &1',systemdt)
30   view-as alert-box.
31 end.
32
33 assign
34   superhdl = ?
35 .
```




OOABL Static

```
SysInfo.cls
1 /*-----*/
2 File      : SysInfo
3 /*-----*/
4 using Progress.Lang.*.
5
6 block-level on error undo, throw.
7
8 class pca.System.SysInfo:
9
10  /* constructor is optional */
11  constructor static SysInfo ( ):
12    /* code here will run ONCE per session -- on first invocation */
13    end constructor.
14
15  /* static method: can be called without new instance */
16  method static public char SystemDateTime().
17    return string(datetime-tz(now)).
18  end. /* SystemDateTime */
19 end class.
20 /* EOF */

run_SuperAsClass.p
1 /*-----*/
2 File      : run_SuperAsClass.p
3 /*-----*/
4
5 /* ***** Definitions ***** */
6 define variable systemdt as character no-undo.
7
8 /* ***** Main Block ***** */
9
10 /* do not need to "NEW" the SysInfo class as it is a 'static' class */
11 assign
12   systemdt = pca.System.SysInfo:SystemDateTime()
13   .
14
15 message subst('System DateTime is: &1',systemdt)
16 view-as alert-box.
17
18 /* nothing to clean-up */
19 /* EOF */
```




Every-day Use Examples



Every-day Use: OpenEdge.Core.Assert

- Easily the most versatile class
- Static methods
 - Covers many every-day use-cases



Every-day Use: OpenEdge.Core.Assert

Procedural Validation

```
1 run TestInputVal
2   ( input '' /* ', 'a', ? */ )
3   no-error.
4
5 /* test for error: check return-value */
6 if (error-status:error) then
7 do:
8   /* fail, leave, return, etc */
9   message "An Error Occurred"
10  skip return-value
11  view-as alert-box.
12 end.
13 else
14 do:
15   message "all went well"
16   view-as alert-box.
17 end.
18
19 procedure TestInputVal
20 :
21   define input parameter inputValue as character no-undo.
22
23   define variable errmsg as character no-undo.
24
25   if (inputValue eq ? or
26       inputValue eq "" ) then
27 do:
28   /* raise some kind of error */
29   assign errmsg = "bad inputvalue parameter".
30 end.
31 else
32 do:
33   /* do stuff here, knowing inputValue has a value */
34 end.
35
36   if (errmsg gt '') then
37     return error errmsg.
38   else
39     return.
40 end. /* TestInputVal */
```

OOABL Assertions

```
1 using OpenEdge.Core.Assert from propath.
2
3 block-level on error undo, throw.
4
5 run AssertInputVal
6   ( input '' /* 'a', ? */ ).
7
8 message "all went well"
9 view-as alert-box.
10
11 catch err as Progress.Lang.Error
12 :
13   message "An Error Occurred"
14   skip err:GetClass():TypeName
15   skip err:GetMessage(1)
16   view-as alert-box.
17
18   delete object err.
19   assign err = ?.
20 end catch.
21
22 procedure AssertInputVal
23 :
24   define input parameter inputValue as character no-undo.
25
26   Assert:NotNullOrEmpty(inputValue).
27
28   /* do stuff here, knowing inputValue has a value */
29   return.
30 end. /* AssertInputVal */
31 /* EOF */
```

Every-day Use: OpenEdge.Core.Session



Procedural Session

```
1 define variable hProc as handle no-undo.
2
3 hProc = session:first-procedure.
4 do while valid-handle(hProc) and hProc:file-name ne <filename>
5     :
6     hProc = hProc:next-sibling.
7 end.
8
```

OOABL Session

```
1
2 using OpenEdge.Core.Assert from propath.
3
4 define variable filehdl as OpenEdge.Core.WidgetHandle no-undo.
5
6 assign
7     filehdl = OpenEdge.Core.Session:GetFirstRunningProc('<filename>')
8     .
9
10 /* Assert */
11 Assert:NotNull(filehdl, '<filename>').
12
13 /* OR throw custom error */
14 if not valid-object(filehdl) then
15     undo, throw new Progress.Lang.AppError('Unable to locate <filename>', 001).
16
```


Every-day Use: OpenEdge.Core.TimeStamp



Every-day Use: OpenEdge.Core.String

Procedural Strings

```
strings_only_proc.p
1 /*-----
2   File       : strings_only_proc.p
3   -----*/
4 &scoped-define delim ","
5 define variable mystring as character
6
7 /* initial string */
8 assign mystring = "this".
9
10 /* concatenation */
11 assign mystring = subst('&1&2&3'
12                        ,mystring
13                        ,{&delim}
14                        ,"theother"
15                        ).
16 /*mystring === "this,that,theother" */
17
18 /* EOF */
```

OOABL Strings

```
strings_only_oop.p
1 /*-----
2   File       : strings_only_oop.p
3   -----*/
4 using OpenEdge.Core.String from propath.
5
6 &scoped-define delim ","
7 define variable mystring as String no-undo.
8
9 assign mystring = new String("this").
10
11 /* concatenate */
12 mystring:Append( "that"      ).
13 mystring:Append( {&delim}    ).
14 mystring:Append( "theother" ).
15 /*mystring:ToString() === "this,that,theother", limits to length ~ 32k */
16 /*mystring:Value      === longchar, full string */
17
18 /* EOF */
```

Every-day Use: OpenEdge.Core.Collections.Array

Procedural Array/Extent (variable len)

OOABL Array

```
strings_proc.p
1  /*-----
2  File       : strings_proc.p
3  -----*/
4  &scoped-define delim ","
5  define variable mystring as character no-undo
6  define variable extstring as character extent no-undo
7  define variable ilen as integer no-undo
8  define variable ii as integer no-undo
9
10 /* initial string */
11 assign mystring = "this".
12
13 /* concatenation */
14 assign mystring = subst('&1&2&3'
15                        ,mystring
16                        ,{&delim}
17                        ,"theother"
18                        ).
19 /*mystring === "this,that,theother" */
20
21 /* make it an array/extent */
22 assign
23     ilen = num-entries(mystring,{&delim})
24     extent(extstring) = ilen /* set length, determinat
25     .
26 do ii = 1 to ilen
27     :
28     assign extstring[ii] = entry(ii,mystring,{&delim}).
29 end.
30 /* extstring === [ "this","that","theother" ] */
31
32 /* EOF */
```

```
strings_oop.p
1  /*-----
2  File       : strings_oop.p
3  -----*/
4  using OpenEdge.Core.String from propath.
5  using OpenEdge.Core.Collections.Array from propath.
6
7  &scoped-define delim ","
8  define variable mystring as String no-undo.
9  define variable extstring as Array no-undo.
10
11 assign mystring = new String("this").
12
13 /* concatenate */
14 mystring:Append( "that" ).
15 mystring:Append( {&delim} ).
16 mystring:Append( "theother" ).
17 /*mystring:ToString() === "this,that,theother", limits to length ~ 32k */
18 /*mystring:Value      === longchar, full string */
19
20 /* make it an Array */
21 assign extstring = String:Split(mystring). /* or Split(mystring,{&delim}) */
22 /* extstring:Size = 3 */
23
24 /* EOF */
```


Every-day Use: OpenEdge.Core.Collections.StringCollection

- A Better Way to build a list of Strings
- Use Iterator to loop entries
- Consider it a temp-table of typed-objects

```
string_collection.p
1 using OpenEdge.Core.Collections.StringCollection from propath.
2
3 define variable str as StringCollection no-undo.
4
5 /* create the collection */
6 assign
7     str = new StringCollection
8     .
9
10 /* add strings to the collection */
11 str:Add("this").
12 str:Add("that").
13 str:Add("theother").
14
15 message str:Size /* 3 */
16 /* does the Collection have this string? */
17 skip str:Contains("that") /* YES */
18 /* create a delimited list */
19 skip OpenEdge.Core.String:Join(strs:ToStringArray(),",") /* "this,that,theother" */
20 view-as alert-box.
21 /* EOF */
```

```
define variable striter as OpenEdge.Core.Collections.IIterator no-undo.
/* get an iterator for the collection */
assign striter = str:Iterator().
/* create a loop using the Iterator */
do while striter:HasNext()
:
message cast(striter:Next(),OpenEdge.Core.String):ToString()
view-as alert-box.
end.
```

Every-day Use: OpenEdge.Net.URI


- String-parsing of URIs can be hit-or-miss
- URI has both Static and Instance methods

```
1 using OpenEdge.Net.URI from propath.  
2  
3 define variable oURI as URI no-undo.  
4  
5 /* use Static invocation to PARSE a URL */  
6 assign  
7     oURI = URI:Parse('http://pugchallenge.org/agenda.html')  
8  
9  
10 /* manipulate QueryString */  
11 oURI:AddQuery('sess','227').  
12  
13 message subst('oURI:Scheme      : &1',oURI:Scheme      )  
14 skip subst('oURI:Host         : &1',oURI:Host         )  
15 skip subst('oURI:Port         : &1',oURI:Port         )  
16 skip subst('oURI:Path         : &1',oURI:Path         )  
17 skip subst('oURI:QueryString: &1',oURI:QueryString)  
18 view-as alert-box.  
19  
20 finally:  
21     /* clean-up */  
22     delete object oURI no-error.  
23     assign  
24         oURI = ?  
25  
26 end finally.  
27 /* EOF */
```

Message (Press HELP to view stack trace)

oURI:Scheme : http
oURI:Host : pugchallenge.org
oURI:Port : ?
oURI:Path : /agenda.html
oURI:QueryString: ?sess=227

OK Help



Every-day Use: OpenEdge.Net.HTTP.*

```
1  /*-----*/
2  File      : example_HTTPget.p
3
4  HTTPClient : who is requesting?
5  URI       : where is the resource the client is requesting?
6             (could be passed in as string parameter)
7
8  HTTPRequest : the request object, payload/data/headers/etc
9  HTTPResponse: the response object, payload/data/headers/etc
10
11  HTTPResponse:Entity -> contents returned (body of html, json, etc.)
12  -----*/
13
14  /* ***** Definitions ***** */
15
16  block-level on error undo, throw.
17
18  using OpenEdge.Core.Assert from propath.
19  using OpenEdge.Core.Assertion.AssertObject from propath.
20
21  /* interfaces: objects are built for us, we just need to know the "type" */
22  define variable oClient as OpenEdge.Net.HTTP.IHttpClient no-undo.
23  define variable oReq   as OpenEdge.Net.HTTP.IHttpRequest no-undo.
24  define variable oRes   as OpenEdge.Net.HTTP.IHttpResponse no-undo.
25
26  /* classes: we will build these objects ourselves. */
27  define variable oUri   as OpenEdge.Net.URI no-undo.
28
29  /* ***** Main Block ***** */
30
31  assign
32  /* Build a URI for the "WHERE" of the resource */
33  oUri = new OpenEdge.Net.URI('http', 'pca2017.thomson.net', 80)
34  oUri:Path = '/'
35
36  /* ClientBuilder does the "NEW" for us. */
37  oClient = OpenEdge.Net.HTTP.ClientBuilder
38           :Build() /* DefaultHTTPClientBuilder */
39           :SetRequestTimeout(10) /* seconds */
40           /* ... other options here ... */
41           :Client /* [I]HTTPClient */
42
43  /* Prepare a Request object for the client to use */
44  oReq = OpenEdge.Net.HTTP.RequestBuilder
45       :Get(oUri) /* DefaultRequestBuilder */
46       :Request /* [I]HttpRequest */
47
```

```
49 assign
50 /* tell the client to execute the request, capture the response */
51 oRes = oClient:Execute(oReq)
52 .
53
54 message subst('Response : &1 &2', oRes:StatusCode, oRes:StatusReason)
55 view-as alert-box.
56
57 /* Response Status Code must be positive */
58 Assert:IsPositive(oRes:StatusCode, 'Response Status Code').
59 Assert:Object:Equals(OpenEdge.Net.HTTP.StatusCodeEnum:GetEnum(oRes:StatusCode), OpenEdge.Net.HTTP.StatusCodeEnum:OK)
60
61 message
62 skip subst('Contents : &1 &2', oRes:ContentType, oRes:ContentLength)
63 skip subst('EntityType: &1', oRes:Entity:GetClass():TypeName)
64 view-as alert-box.
65
66 if (type-of (oRes:Entity, 'OpenEdge.Core.String')) then
67 do:
68 message cast(oRes:Entity, OpenEdge.Core.String):ToString()
69 view-as alert-box.
70 end.
71
72 catch err as Progress.Lang.Error
73 :
74 message subst('Error: &1', err:GetMessage(1))
75 skip err:CallStack
76 view-as alert-box.
77
78 delete object err.
79 assign err = ?.
80 end catch.
81
82 finally:
83 delete object oUri no-error.
84 delete object oClient no-error.
85 delete object oReq no-error.
86 delete object oRes no-error.
87
88 assign
89 oUri = ?
90 oClient = ?
91 oReq = ?
92 oRes = ?
93 .
94 end finally.
```

Message (Press HELP to view stack trace)

Response : 200 OK

OK Help

Message (Press HELP to view stack trace)

Contents : text/html 117
EntityType: OpenEdge.Core.String


OK Help

Message (Press HELP to view stack trace)

```
<html>
<head>
<title>pca2017.thomson.net</title>
</head>
<body>
<h1>pca2017.thomson.net</h1>
</body>
</html>
```


OK Help

What to Expect




What to Expect

- Adopt **PDSOE** for your IDE
 - The auto-complete and class-viewer are invaluable
 - At minimum, use an IDE that performs color-coding and syntax checks
- Confusion – Move at a comfortable pace for you
 - Start simple
 - Move deeper into OOP as your understanding increases
 - Learn using good examples
- Reference libraries will change between versions
 - Acquire the source and actually look at it
 - Compare the source between versions
 - Compilations will let you know if things are *broken*, you will want to confirm the behavior hasn't changed beyond your expectations



Next Steps

- Create your own custom override classes
 - Explore use of overloads to simplify call signature
- Interfaces, Abstract classes
- JSON
 - Progress.JSON.*
- Enumerators
 - Progress.Lang.Enum
 - OpenEdge.Core.*Enum
- Runtime Application Framework
 - CCS-style



Next Steps

- Next-level OOP design recommendations
 - CCS Samples: https://github.com/consultingwerk/CCS_Samples
 - IOC, Dependency Injection: <https://github.com/PeterJudge-PSC/InjectABL>
 - AutoEdge, The Factory: <https://github.com/PeterJudge-PSC/autoedgethefactory>
- (2016) 806: OO-Oh, Mike Fechner, Consultingwerk, Ltd.
 - <http://pugchallenge.org/downloads2016/806 - OO-Oh.pdf>
- (2018) 326: Building Great Interfaces with OOABL, Mike Fechner, Consultingwerk, Ltd.
 - http://pugchallenge.org/downloads2018/Fechner_OOABL.PDF
- Sessions by Peter Judge
 - excellent resources – couldn't find links :(
- OE Dev Tools Source
 - Progress Community Website
 - https://community.progress.com/community_groups/openedge_general/w/openedgegeneral/tags/development_5F00_tools





Questions?

PUGCHALLENGE  **EXCHANGE**
AMERICAS